

Advanced Finite Elements

ME EN 7540

Plastic Bending of a Clamped Beam

Spring 2006

Example 1

In this example, we will investigate the behavior of a cantilever beam under larger deflection. When the model undergoes larger deflection, the basic analysis that is often used in ANSYS is no longer sufficient. The load has to be broken down into small steps (load steps) and the stiffness matrix is then updated each time using the result from the previous load step. Theoretical details on geometric nonlinearity can be found in the class handout, hence, we will only focus on how to perform such analysis using ANSYS here. The model to be analyzed in this example is illustrated in Figure 1.




Figure 1. Cantilever beam sketch

Material Properties	Geometric Properties	Loading
$E = 1 \times 10^6$ psi $\sigma_y = 5.8 \times 10^6$ psi	$\ell = 100$ in $b = 2$ in $h = 2.5$ in	$P = 400$ lb

Input Listing

```
/PREP7
/TITLE, GEOMETRICALLY NONLINEAR ANALYSIS OF A CANTILEVER BEAM

ET,1,42
!this option allows for the analysis of plane stress with thickness
KEYOPT,1,3,3
R,1,2

!material properties
MP,EX,1,1E6
MP,PRXY,,0.3

!model of the structure
K,1,0,0
K,2,100,0
K,3,100,2.5
K,4,0,2.5
L,1,2,100
L,2,3,2
L,3,4,100
L,4,1,2
A,1,2,3,4

!mesh the areas
AMESH,ALL

!apply constraints to the left end
NSEL,S,LOC,X,0,0
D,ALL,ALL,0

!apply load to the top-right corner
NSEL,S,LOC,X,100,100
NSEL,R,LOC,Y,2.5,2.5
F,ALL,FY,-400
NSEL,ALL

FINISH

/SOLU
ANTYPE,0

!here we allow ANSYS to control the step size
AUTOTS,0

!set the option so that this analysis is for large deformation
NLGEOM,1

!set number of substep to 100
NSUBST,100

solve
fini
/post1
pldisp,1
fini

!here we run the analysis again without using the large deformation feature of ANSYS
/SOLU
ANTYPE,0
NLGEOM,0
NSUBST,0
SOLVE
FINI
/POST1
PLDISP,1
FINI
```

Results

The plot of the deformations of the cantilever beam is shown in Figure 2 and 3 for nonlinear and linear analysis, respectively.




Figure 2 Deformation using nonlinear analysis.




Figure 3 Deformation using linear analysis.

Example 2

A wide-flanged I-beam of length l , with clamped ends, is uniformly loaded as shown. Investigate the behavior of the beam at load w_1 when pronounced plastic yielding has occurred. The beam's cross-section is shown in Figure 3.




Figure 4 Clamped I-Beam Problem Sketch.

Material Properties	Geometric Properties	Loading
$E = 29 \times 10^6$ psi	$l = 144$ in	$w_1 = 9039$ lb/in
$E_T = 5.8 \times 10^6$ psi	$b = 10$ in	
$\sigma_y = 38,000$ psi	$h = 10.6$ in	
	$t_f = 0.9415$ in	
	$t_w = 0.0001$ in	

Analysis Assumptions and Modeling Notes

The beam cross-section is modeled as an idealized section to compare with the assumptions of the analytical solution. The loading is assumed to be applied through the centroid of the element cross-section (the neutral axis). Only half the beam is modeled, taking advantage of symmetry. Classical bilinear kinematic hardening behavior is used.

Input Listing

```
/PREP7
MP,PRXY,,0.3
/TITLE, PLASTIC BENDING OF A CLAMPED I-BEAM
ANTYPE,STATIC

! PLASTIC BEAM WITH CENTROID AT NODES
ET,1,BEAM24,,,1

! BEAM SECTION PROPERTIES
R,1,0,0,0,10,0,.9415
RMORE,5,0,0,5,10.6,.0001
RMORE,0,10.6,0,10,10.6,.9415

! BILINEAR KINEMATIC HARDENING BEHAVIOR
MP,EX,1,29E6
TB,BKIN,1,1
TBTEMP,0.0

! YIELD STRESSES AND TANGENT MODULUS
TBDATA,1,38000,5.8E6
N,1
N,10,72
FILL
N,100,,,1

E,1,2,100
*REPEAT,9,1,1

! FIX ONE END
D,1,ALL

! SYMMETRIC MID-SPAN B.C.
D,10,ROTY,,,,UY,ROTX,ROTZ
FINISH

/SOLU
SOLCONTROL,0
SFBEAM,1,1,PRES,9039
*REPEAT,9,1
SOLVE
FINI

/POST1
etable,MI,smisc,5
etable,MJ,smisc,11
pretab,MI,MJ
prdisp
fini
```

Results

Refer to nodes and elements in Figure 4.

Element Solution

Element	Moment about y at node I (lb-in)	Moment about y at node J (lb-in)
1	0.14972E+08	0.10055E+08
2	0.10055E+08	0.57162E+07
3	0.57099E+07	0.19499E+07
4	0.19559E+07	-0.12258E+07
5	-0.12283E+07	-0.38318E+07
6	-0.38288E+07	-0.58537E+07
7	-0.58537E+07	-0.73000E+07
8	-0.73000E+07	-0.81677E+07
9	-0.81677E+07	-0.84570E+07

Nodal Displacement in global coordinate

NODE	UX (in)	UY (in)	UZ (in)	ROTX (rad)	ROTY (rad)	ROTZ (rad)
1	0	0	0	0	0	0
2	-3.04E-17	8.60E-17	-0.10733	6.12E-17	2.47E-02	1.14E-17
3	-4.54E-17	1.37E-16	-0.36293	1.06E-16	3.73E-02	1.21E-18
4	-4.75E-17	1.30E-16	-0.68135	1.32E-16	4.13E-02	-1.84E-18
5	-4.76E-17	1.13E-16	-1.0135	1.44E-16	4.15E-02	-2.42E-18
6	-4.63E-17	9.24E-17	-1.3405	1.48E-16	4.01E-02	-2.66E-18
7	-4.02E-17	7.21E-17	-1.6457	1.36E-16	3.53E-02	-2.46E-18
8	-3.00E-17	4.16E-17	-1.8938	1.08E-16	2.61E-02	-1.16E-18
9	-1.00E-17	2.35E-17	-2.0548	6.21E-17	1.38E-02	-5.14E-18
10	-1.69E-17	0	-2.1104	0	0	0